

EXCEL マクロを使ったアンケート収集方法

山 守 一 徳[†]

A Collection Method of Questionnaires Using EXCEL Macro

Kazunori YAMAMORI

要 旨

大学ではアンケートを取らなければならない活動が増え、そのための労力をかなり要している。特に、集計する作業は手間が掛かる。そこで、EXCEL ファイルを受け渡しすることによって、集計を容易にしたいと考えた。実際に、学部の全教員から研究実績アンケートを取るためと、教職に就いた卒業生からアンケートを取るための2種類のアンケートにおいてEXCEL ファイルを受け渡しする方法で実施してみたので、その方法について述べる。

1. はじめに

学部の全教員から、毎年研究実績を把握するために、論文数や学会発表数などの数値を収集する必要がある。この研究実績アンケートは件数の収集に合わせ、賞をいただいたものなど、特に優れた内容と思われる研究実績については、論文タイトル等の情報も収集しなければならない。この集めた情報から、文部科学省へ提出される現況調査表などの基礎データを作成する訳である。

また、大学時代で教えられた内容が役に立ったかを就職後に現場で働き出した状況で、卒業生へ問い合わせを行うアンケートがあり、この卒業生アンケートの収集作業も相当な手間の掛かる作業となっている。

どちらも以前は紙でアンケート用紙を配布し、回答されたアンケート用紙を回収し、その中の記載を教員が手作業で集計することをしていた。この集計作業を簡単にするために、EXCEL を用いて実施した。

2. アンケートを集計する方法

2.1 アンケート用紙の形式

ここでは、研究実績を集計する作業を例にして、実現した方法を示すことにする。研究実績を尋ねるアンケートでは、発表論文数などの数値を尋ねる箇所と、受賞した賞などの文字記述を必要とする箇所がある。EXCEL では、図1のような欄を用意し、各欄の中に数値か文字列を記入していただく必要がある。図1で示すように、欄のセル位置が固定されていることが重要なポイントであり、後半に文字列を入力する欄があるが、この欄の個数は、多めの個数分の入力欄を用意することで、次の質問の開始欄が下方向へ移動しないようにすることも肝心である。このファイルを全教員へ送り、中身を記入して

[†] 三重大学教育学部

いただいた後に、Moodle の中へ記入済みファイルを提出していただいた。Moodle を使うのは、学部の全教員相手なので、ログインアカウントを全員保有しているためである。さらに、提出履歴が残り、未提出者が簡単に確認できるからである。

2. 2 数値集計するための EXCEL マクロ

数値集計用の EXCEL ファイルを図 2 に示す。このファイルは、アンケートで配布した EXCEL ファイルとは別に用意する。

上部に表示されている件数集計ボタンを押すと EXCEL マクロが実行されて件数が表示されてくる。その EXCEL マクロを付録 1 に示す。

特定のフォルダの中に図 1 の EXCEL ファイルを全部置き、そのファイルの中の特定の欄の数値を合計するプログラムになっている。件数集計ボタンの左側には、集計に使われた EXCEL ファイルの数が表示され、集計した日付も表示されるようになっている。

2. 3 文字列記述を収集するための EXCEL マクロ

図 3 に文字列で記述された回答部分を収集してくるための EXCEL ファイルを示す。図 2 のファイルとは同一ファイルであり、シートを別にして生成した。

図 3 の例では連携実績集計のボタンを押すと、7 行目以降に記述文が表示されるようになっている。

図 2 の場合と同様に、特定フォルダの中の EXCEL ファイルをすべて読み込み、特定の欄に記述してある文章を収集してくるようになっている。例えば図 1 の下部の回答文の記述された行を集めてくる動

図 1 研究実績アンケート用紙の EXCEL ファイル

図 2 数値集計のための EXCEL ファイル

きになっている。工夫点は、回答用紙には氏名を書く欄が設けてあり、その欄の文字列を図3の回答者の欄にコピーしてきている。これによって、EXCEL ファイル読み込み途中に、どのファイルで読み込み異常が起きたかを知ることができる。このEXCEL マクロを付録2に示す

3. アンケートを発信する方法

3.1 卒業生アンケートの形式

図4に卒業生アンケートのEXCEL ファイルを示す。回答は、ほとんどが選択肢の中から選び、その数値を欄に記載するようになっている。最下部には自由記述の設問が数個並んでいる。EXCEL ファイルで回答していただくと、この自由記述の文字列を集計者が手打ち入力することなく集めることができるという利点がある。

卒業生アンケートの難点は、これらのアンケートを卒業生に配布する方法である。在学中ならば、大学から支給されるメールアドレスを使って配布し、Moodle を使って回収することが容易であるが、卒業後にアンケート発信するため彼らの卒業後に使えるメールアドレスを収集しなければならない。そこで、卒業後間もないうちに、メールアドレスの収集に努めた。そして、そのメールアドレスへEXCEL ファイルを添付してアンケート発信を行った。

3.2 アンケートの発信管理の方法

卒業生のメールアドレスへEXCEL ファイルを添付して発信するのであるが、そのメールの本文や件名、発信者のメールアドレスなどは共通なので、それらの情報をシートで管理し、その中の記載情報を使ってメール発信する。そのシートの例を図5に示す。このようにすると、アンケートの回答が戻って来ない卒業生に、催促メールを出す時には、メール本文の欄やメール件名の欄を書き替えてメール発信することができる。

卒業生のメールアドレスの管理は、図6のように、送信履歴、送信選択フラグ、所属、学籍番号、氏




図3 記述収集のためのEXCEL マクロ




図4 卒業生アンケート用紙のEXCEL ファイル

名、メールアドレス、就職先、正規／非正規の欄を用意して管理する。送信選択フラグを1に指定した卒業生へメール発信され、その時の送信日時と送信順番が送信履歴の欄へ上書きで記載される。そのEXCELマクロのプログラムを付録3に示す。今回のアンケートは卒業生の中でも教職に就職した学生のみであるので、就職先の欄が学校名である行のみに、送信選択フラグを1にセットし、マクロを起動させる。

	A	B
1	発信者メールアドレス	xxxxxx@edu.mie-u.ac.jp
2		
3	送信方法(2=SMTP)	2
4	SMTPサーバー	xxxxxx.edu.mie-u.ac.jp
5	ポート番号	25
6		
7		
8	メール件名	三重大学教育学部からのお願い 卒業生・修了生の皆様へ
9	メール本文	三重大学教育学部では、カリキュラム構成 や各科目の授業内容の改善・改良を重ね、 より良いものとしていくために、
10	添付ファイルのフルパスf¥自己点検¥アドレス¥アンケート用紙.xls	
11		

図5 メール発信内容のシート

回答が戻ってきた時に、送信選択フラグの欄に回収済と書き込みを行う。このようにすると回答が戻って来ない卒業生は送信選択フラグが1のままとなるので、催促メールを出すべき相手をこれで把握することができる。

	A	B	C	D	E	F	G	H	I
1	送信履歴	送信選択フラグ	所属	学籍番号	氏名	連絡先	就職先	正規/非正規	
2			情報	211***	山A	abc@gmail.com	三重大学		
3	2015/08/03 9:30:10 + 1	1	情報	211***	山B	abc@yahoo.co.jp	津市立北小学校	非正規	
4	2015/07/28 13:11:26 + 1	回収済	情報	211***	山C	def@gmail.com	松阪市立鈴中学校	正規	
5			情報	211***	山D	def@docomo.ne.jp	教育株式会社		
6	2015/08/03 9:30:10 + 2	1	情報	211***	山E	ghi@gmail.com	亀山市立鶴川中学校	正規	

図6 卒業生のメールアドレスと送信履歴管理のシート

3. 3 集計方法

回答の返ってきたメールの中に添付されたEXCELファイルを取り出し、特定のフォルダの下へ保存し、その後、2章で述べた方法と同じようなEXCELマクロを使って集計を行う。アンケートの回答となるEXCELファイルの中には氏名が入っておらず、EXCELファイルだけでは誰の回答であるかわからないが、返信メールの本文の中に氏名を記載していただき、誰から返答があったのかを管理することにした。未回答者へ催促メールを出すことをメール本文の中に記載していたため、回答者は氏名を書いてきてくれた。

4. 実施状況

4. 1 研究実績アンケートの実施状況

Moodleを利用したことがない教員は、アンケートをMoodle内へ提出するのに苦労していた。最後まで操作方法がわからない場合は、メールにEXCELファイルを添付して送っていただいた。91件を回収し、最終的には2名がEXCELファイルでの提出ができなかった。また、EXCELマクロの集計するプログラムを動かした時に読み込みエラーを起こしたファイルが1つ存在した。セル位置を壊してしまったものと思われる。

4. 2 卒業生アンケートの実施状況

学生の所属した講座によって、卒業生のメールアドレスの把握度合が異なり、普段、メール連絡を行っていないと思われる講座では、メールアドレスの収集が困難であった。結局、メールアドレスが把握できない卒業生に対しては、通常郵便で紙によるアンケートを行った。結局、教職になった卒業生で紙によるアンケートを行ったのは 14 通だった。メールアドレスが把握できた卒業生には、EXCEL ファイルを使って、メール添付によるアンケート回収を目指したが、EXCEL ソフトを持っていない卒業生がいると思われ、最終的には添付ファイルを使わずメール本文で質問や回答もやり取りすることも行った。また、多くの卒業生は、PC メールアドレスを普段使っていない様子であり、ケータイのメールアドレスを見ている様子である。そのため、PC メールアドレスにメールが届いていることに気付いていないのか、催促メールを 4 度、5 度と出す必要があった。結局、このシステムによる私からの催促メールだけでなく、講座の先生からも催促を出していただいた。メールでアンケート発信したのは 104 名、そのうちメールでアンケート回収できたのは 56 名だった。郵便でアンケート発信したのは 14 名、そのうち返信をいただいたのは 9 名だった。どちらも例年より回収率は良かった。

5. まとめ

EXCEL マクロを使うアンケート集計は非常に簡単に集計ができる。そのための前準備が必要であるが、付録のプログラムを参考にされて、活用されることをお勧めする。

なお、統一アカウントで最後にログインした時点でのパスワードを使って卒業後もログインできる「卒業生 Moodle」(<http://portal.mie-u.ac.jp/commu/>) が三重大学には存在している。知名度が上がれば、こちらを利用することも考えられる。

今後の課題

卒業生のメールアドレスを収集するまでが講座によって大変な作業となった。そのため、在学中にメールアドレスを収集するように努めたい。大学から支給されるメールアドレスは、卒業後に使えなくなるため、その前までに、EXCEL ファイルをやり取りできるフリーのメールアドレスを取得するように 4 年生へ勧めたい。そして、卒業前までに各自の PC 環境を整えておくようにしておけば、卒業生とのメールによるアンケートは容易に実施できると思われる。

参考文献

- (1) 七條達弘「やさしくわかる ExcelVBA プログラミング第 5 版」ソフトバンククリエイティブ (2013 年 9 月発行)
- (2) 「Excel VBA 入門講座へようこそ!!」<http://excelvba.pc-users.net> (2015 年 10 月現在)

付録 1 数値集計するプログラム

```

Sub 件数集計_Click()
    Application.ScreenUpdating = False      '画面のちらつきを防止する
    'プログラムに必要な変数を宣言する

    Dim path As String                      'ファイルパス
    Dim filename As String                  'ファイル名
    Dim answerWB As Workbook                '回答ワークブック名
    Dim answerNo1 As Variant                '回答ワークブック (集計対象になる回答 1 番目)
    (略)
    Dim answerNo7 As Variant                '回答ワークブック (集計対象になる回答 7 番目)
    Dim answerName As String                '回答者名
    Dim answerCellRowNo As Long             '回答ワークブックの現在集計している行数
    Dim totalWB As Workbook                 '集計ワークブック名
    Dim totalNo1(9) As Integer              '集計ワークブック (集計対象になる回答 1 番目)
    (略)
    Dim totalNo7 As Integer                 '集計ワークブック (集計対象になる回答 7 番目)
    Dim totalCnt As Long                    '集計ワークブック (現在の回答数)
    Dim totalCellRowNo As Long              '集計ワークブックの現在集計している行数
    Dim numPeople As Long                   '集計人数
    Dim i As Integer

    '回答 Excel ファイルを保存しているフォルダを指定する ("'"内を修正すること)
    path = "F:\自己点検\自己点検個別アンケート\提出物\"
    numPeople = 0      '集計人数初期化
    '集計ブックの情報を記録する
    Set totalWB = ThisWorkbook
    '集計ブックの初期化处理
    With totalWB.Worksheets("件数集計")
        .Range("B3") = Date      '集計ブックの更新日を作成する
    End With
    filename = Dir(path)      '保存したフォルダを指定して、ファイル名を 1 つ取り出す
    totalCellRowNo = 41      '集計ブックの書き込み先の先頭行
    'フォルダに含まれる Excel ファイルをすべて取得する
    '1 つもファイルがなければ、即時終了する。
    Do While Len(filename) > 0
        '拡張子.xlsx であるときのみ、集計処理を行う
        If Right(filename, 5) = ".xlsx" Then
            '読み込んだ回答ブックを開いて、回答ブックの情報を記録する
            Set answerWB = Workbooks.Open(path & filename)
            '選択した回答ブックから処理開始行を選択する
            '処理開始位置を E10 とする

```

```

answerWB.Activate
answerName = answerWB.Worksheets("Sheet1").Cells(6, 3).Value
'回答ブックの回答を取得する
answerNo1 = answerWB.Worksheets("Sheet1").Range("E10:E18")
(略)
answerNo7 = answerWB.Worksheets("Sheet1").Range("G39")
totalWB.Activate
totalWB.Worksheets("件数集計").Cells(totalCellRowNo, 1).Activate
'回答ブックの回答名を集計ブックに挿入
ActiveCell.Value = answerName
totalCellRowNo = totalCellRowNo + 1
For i = 1 To 9
    totalNo1(i - 1) = totalNo1(i - 1) + answerNo1(i, 1)
Next i
(略)
totalNo7 = totalNo7 + answerNo7
'回答ブックを閉じる
answerWB.Close
'集計人数を増加
numPeople = numPeople + 1
End If
'次のファイルを読む
filename = Dir()
Loop
totalWB.Activate
totalWB.Worksheets(" 件 数 集 計 ").Range("E10:E18") =
WorksheetFunction.Transpose(totalNo1)
(略)
totalWB.Worksheets("件数集計").Range("G39") = totalNo7
'集計ブックの集計人数を更新する
totalWB.Activate
totalWB.Worksheets("件数集計").Range("B4") = numPeople
totalWB.Worksheets("件数集計").Range("B4").Select
'作業領域の開放
Set answerWB = Nothing
Set totalWB = Nothing
'画面のちらつきを防止する
Application.ScreenUpdating = True
End Sub

```

付録2 文字列記述を収集するプログラム

```

Sub 連携実績集計_Click()
    Application.ScreenUpdating = False      '画面のちらつきを防止する
    'プログラムに必要な変数を宣言する

    Dim path As String                      'ファイルパス
    Dim filename As String                  'ファイル名
    Dim answerWB As Workbook               '回答ワークブック名
    Dim answerQNo As String                 '回答ワークブック (集計対象になる回答番号)
    Dim answerVal As String                 '回答ワークブック (質問に対する回答)
    Dim answerName As String               '回答者名
    Dim answerCellRowNo As Long            '回答ワークブックの現在集計している行数
    Dim totalWB As Workbook                '集計ワークブック名
    Dim totalQNo As String                  '集計ワークブック (集計対象になる回答番号)
    Dim totalVal As String                  '集計ワークブック (質問に対する回答)
    Dim totalCnt As Long                   '集計ワークブック (現在の回答数)
    Dim totalCellRowNo As Long             '集計ワークブックの現在集計している行数
    Dim numPeople As Long                   '集計人数
    '回答 Excel ファイルを保存しているフォルダを指定する ("内を修正すること)
    path = "F:\自己点検\自己点検個別アンケート\提出物"
    numPeople = 0      '集計人数初期化
    Set totalWB = ThisWorkbook      '集計ブックの情報を記録する
    With totalWB.Worksheets("連携実績集計")      '集計ブックの初期化処理
        .Range("B3") = Date      '集計ブックの更新日を作成する
        .Range("C7").Select      '以前の集計人数が存在すれば、削除
        .Range(Selection, Selection.End(xlDown)).Select
        Selection.ClearContents
    End With
    filename = Dir(path)      '保存したフォルダを指定して、ファイル名を1つ取り出す
    totalCellRowNo = 7      '集計ブックの書き込み先の先頭行
    'フォルダに含まれる Excel ファイルをすべて取得する '1つもファイルがなければ、終了
    Do While Len(filename) > 0
        If Right(filename, 5) = ".xlsx" Then      '拡張子.xlsx であるときのみ、集計
            '読み込んだ回答ブックを開いて、回答ブックの情報を記録する
            Set answerWB = Workbooks.Open(path & filename)
            '選択した回答ブックから処理開始行を選択する
            answerCellRowNo = 43      '処理開始位置を 43 行目とする
            answerWB.Activate
            answerWB.Worksheets("Sheet1").Activate
            answerWB.Worksheets("Sheet1").Cells(answerCellRowNo, 2).Activate
            answerQNo = ActiveCell.Value '回答ブックの回答番号、回答を取得する
            answerVal = ActiveCell.Offset(0, 1).Value
        End If
    End While
End Sub

```



```

If answerQNo = "なし" Or answerQNo = "特になし" Then
    answerQNo = ""
End If
answerName = answerWB.Worksheets("Sheet1").Cells(6, 3).Value
'回答ブックの回答がなければ、ループを抜ける
Do Until ((answerQNo = "") And (answerVal = ""))
    totalWB.Activate
    totalWB.Worksheets("連携実績集計").Cells(totalCellRowNo, 1).Activate
    '回答ブックのデータを集計ブックに挿入
    ActiveCell.Value = answerQNo
    ActiveCell.Offset(0, 1) = answerVal
    ActiveCell.Offset(0, 2) = answerName
    totalCellRowNo = totalCellRowNo + 1
    '次の質問のための準備    'アクティブセルを下方に 1 移動する
    answerCellRowNo = answerCellRowNo + 1
    answerWB.Activate
    answerWB.Worksheets("Sheet1").Cells(answerCellRowNo, 2).Activate
    '回答ブックの回答番号、回答を取得する
    answerQNo = ActiveCell.Value
    answerVal = ActiveCell.Offset(0, 1).Value
Loop
answerWB.Close    '回答ブックを閉じる
numPeople = numPeople + 1    '集計人数を増加
End If
filename = Dir()    '次のファイルを読む
Loop
totalWB.Activate
totalWB.Worksheets("連携実績集計").Cells(totalCellRowNo, 1).Activate
'集計ブックの最後に空を挿入 (ゴミが残っていた時に空行で最後を示すため)
ActiveCell.Offset(0, 1) = ""
ActiveCell.Offset(0, 2) = ""
ActiveCell.Offset(0, 3) = ""
'集計ブックの集計人数を更新する
totalWB.Activate
totalWB.Worksheets("連携実績集計").Range("B4") = numPeople
totalWB.Worksheets("連携実績集計").Range("B4").Select
Set answerWB = Nothing    '作業領域の開放
Set totalWB = Nothing
Application.ScreenUpdating = True    '画面のちらつきを防止する
End Sub

```

付録3 メール発信管理するプログラム

```

Sub sendmail()
  Dim x, n, count As Integer
  Set objMail = CreateObject("CDO.Message")
  Worksheets("setting").Select
  With Sheets("setting")
    objMail.Subject = .Range("B8") 'メール件名
    MailBody = .Range("B9") '本文
    ' 本文の改行コードの確認 ' Lf のみの場合 Cr+Lf に変換
    strBody = Replace(MailBody, vbLf, vbCrLf)
    ' 上記で元が Cr+Lf の場合 Cr+Cr+Lf になるので Cr+Lf に戻す
    MailBody = Replace(strBody, vbCr & vbCrLf, vbCrLf)
    objMail.TextBody = MailBody '本文
    objMail.AddAttachment .Range("B10") '添付ファイル
    objMail.From = .Range("B1") '発信者メールアドレス
    objMail.TextBodyPart.Charset = "ISO-2022-JP" 'エンコード
    objMail.Configuration.Fields.Item("http://schemas.microsoft.com/cdo/configuration/n/sendusing") = .Range("B3") '送信方法
    objMail.Configuration.Fields.Item("http://schemas.microsoft.com/cdo/configuration/n/smtpserver") = .Range("B4") 'SMTP サーバー
    objMail.Configuration.Fields.Item("http://schemas.microsoft.com/cdo/configuration/n/smtpserverport") = .Range("B5") 'ポート番号
  End With
  Worksheets("list").Select
  With Sheets("list")
    x = 1
    count = 1
    n = Cells(Rows.count, "F").End(xlUp).Row
    Do While x <= n '行の最後まで
      If (Cells(x, 2).Value = 1) Then '送信すべき行か
        objMail.to = Cells(x, 6).Value '送信先
        Debug.Print objMail.to
        objMail.Configuration.Fields.Update
        objMail.Send
        Cells(x, 1).Value = Now & " " & count '送信記録
        count = count + 1
      End If
      x = x + 1
    Loop
  End With
End Sub

```